
1 [bookmark: _Toc122434485][bookmark: _Toc128145554]Overview
This document lists all the changes needed to correct issues in the ATS iwd-TTCN3-B2022-09_D22wk49 related to the title of this CR.
Contact:	Shaun Harry
	shaun.harry@keysight.com 	

2 [bookmark: _Toc122434488][bookmark: _Toc295288959][bookmark: _Toc128145555]Corrections required
2. [bookmark: _Toc106975634][bookmark: _Toc109680267][bookmark: _Toc122623878][bookmark: _Toc122434493][bookmark: _Toc295288970][bookmark: _Toc325725665][bookmark: _Toc128145556]Change 1
	Function name
	TC_11_3_5_NR5GC

	Reason for change
	The guard timer is not sufficient to execute all the test steps.

	Summary of change
	Increased the guard timer to 900s

	TTCN module
	NR5GC_Testsuite

	MCC160 Comment
	



Before Change
	  testcase TC_11_3_5_NR5GC() runs on MTC_NR5GC system SYSTEM_NR5GC {
    // @purpose
    //   UAC / Access Identity 1 / New cell not in the country of its HPLMN/EHPLMN 0% access probability / MPS indicator / HPLMN/0%/100% accessibility AC5 / MMTEL-Video call
    var NR5GC_PTC        v_NR5GC      := null;
    var IMS_PTC          v_IMS1       := null;
    var IMS_PTC          v_IMS2       := null;
  
    timer t_GuardTimer := int2float(300);
  
    v_NR5GC := NR5GC_PTC.create("NR5GC") alive;
    v_IMS1 := IMS_PTC.create("IMS1") alive;
  
    f_MTC_ConnectPTCs_NR5GC(system, v_NR5GC, v_IMS1, v_IMS2);
  
    v_NR5GC.start(f_TC_11_3_5_NR5GC());
    v_IMS1.start(f_TC_11_3_5_NR5GC_IMS1());
  
    t_GuardTimer.start;
  
    f_MTC_MainLoop_NR5GC(t_GuardTimer);
  }  



After Change
	   testcase TC_11_3_5_NR5GC() runs on MTC_NR5GC system SYSTEM_NR5GC {
    // @purpose
    //   UAC / Access Identity 1 / New cell not in the country of its HPLMN/EHPLMN 0% access probability / MPS indicator / HPLMN/0%/100% accessibility AC5 / MMTEL-Video call
    var NR5GC_PTC        v_NR5GC      := null;
    var IMS_PTC          v_IMS1       := null;
    var IMS_PTC          v_IMS2       := null;
  
    timer t_GuardTimer := int2float(900);
  
    v_NR5GC := NR5GC_PTC.create("NR5GC") alive;
    v_IMS1 := IMS_PTC.create("IMS1") alive;
  
    f_MTC_ConnectPTCs_NR5GC(system, v_NR5GC, v_IMS1, v_IMS2);
  
    v_NR5GC.start(f_TC_11_3_5_NR5GC());
    v_IMS1.start(f_TC_11_3_5_NR5GC_IMS1());
  
    t_GuardTimer.start;
  
    f_MTC_MainLoop_NR5GC(t_GuardTimer);
  } 




2. [bookmark: _Toc128145557]Change 2
	Function name
	fl_TC_11_3_5_TestBody

	Reason for change
	1. If the REGISTRATION REQ is sent with registration type ‘mobility’, the current TTCN implementation at steps 2Ca1 and 11Fa1 would incorrectly fail the check with cause "Initial Registration Req included non-cleartext IEs", as the current implementation first checks registration type to be “Initial_Secure”.
2. At step 10Fa8, since the SRB2 has already been established, the PDU SESSION ESTABLISHMENT procedure will use SRB2
3. The timer t_Wait at step 10Eb3 is not stopped if the rrcSetupReq is received at this step, which would cause an unexpected failure when this timer expires
4. At step 10H, the IMS PTC would be stuck inside the function f_IMS_OptionalDeregistrationOrC24Reinvite() (called from f_IMS_IpcanReleaseWithOptionalDeregistration () with p_WaitForIpcanRelease = true) waiting for a trigger from NR side to indicate the RRC connection is released, before it can handle the de-REGISTER REQ from the UE.

	Summary of change
	1. Changed the TTCN implementation to call the individual checks based on the registration type by using select case method.
2. Changes the SRB type from SRB1 to SRB2 at this step
3. Timer t_Wait is stopped when rrcSetupRequest is received.
4. Added the coordination trigger to tell IMS PTC to stop waiting for IMS de-REGISTER

	TTCN module
	UAC_NR5GC.ttcn

	MCC160 Comment
	



Before Change
	  function fl_TC_11_3_5_TestBody() runs on NR5GC_PTC
  {
…
…
…
    //@siclog "Steps 2Ca1-2Ca15" siclog@
    //IF 5GS registration type is set as Initial Registration in step 2B, THEN Steps 5 to 19 of the generic test procedure in TS 38.508-1 Table 4.5.2.2-2 are performed without RRC release.on NR Cell 12
    // @sic R5-223436 sic@
    if (f_Check_NG_RegistrationReqMsg (v_GMM_MobilityInfo, v_ReceivedMsg.Pdu, Initial_Secure)) {
    f_NR5GC_RRC_Idle_Steps5_20 (v_GMM_MobilityInfo, nr_Cell12, v_ReceivedMsg, TESTMode_OFF, noRrcConnectionRelease, Initial_Secure);
    } else{
        if (f_Check_NG_RegistrationReqMsg (v_GMM_MobilityInfo, v_ReceivedMsg.Pdu, Mobility)) {
        //@siclog "2Cb1-2Cb2" siclog@
        //IF 5GS registration type is set as Mobility Registration in step 2B, THEN Steps 4 to 5 of the generic test procedure in TS 38.508-1 Table 4.9.5.2.2-1 are performed without RRC release.on NR Cell 12.
        // @sic R5-223436 sic@
          f_NR5GC_MobilityRegistration_Steps4_6(nr_Cell12, v_GMM_MobilityInfo, -, -, noRrcConnectionRelease);
        }
    }
    

    //@siclog "Steps 3 - 10" siclog@
    //Void @sic R5-223436 sic@
    //@siclog "Steps 10A" siclog@
    // @sic R5-223436 sic@
    f_NR_RRCRelease(nr_Cell12);
    //@siclog "Steps 10B" siclog@
    // @sic R5-223436 sic@
    f_UT_RequestPDUSession(UT, f_NR5GC_MobileInfo_GetNoOfSessions()+1, tsc_DNNValue);
    //@siclog "Steps 10C" siclog@
    // @sic R5-223436 sic@
    t_WaitTime.start(21.0);
    alt {
        //Check: Does the UE transmit an RRCSetupRequest message within 21s on NR Cell 12?
        [] SRB.receive(car_NR_SRB0_RrcPdu_IND(nr_Cell12, cr_38508_RRCSetupRequest(?, ?)))
        {
           t_WaitTime.stop;
           f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 10C");
        }
        [] t_WaitTime.timeout {
           f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 10C");
        }
    }
    //@siclog "Steps 10D" siclog@
    // @sic R5-223436 sic@
    //SS changes SIB1 of NR Cell 12 according to Table 11.3.5.3.3-2 and sends Short Message on PDCCH using P-RNTI.
    f_NR_CellInfo_SetSIB1_UacBarringInfo(nr_Cell12, omit);
    f_NR_ModifySysinfo(nr_Cell12);
    //@siclog "Steps 10E" siclog@
    // @sic R5-223436 sic@
    t_WaitTime.start(28.0);
    alt {
        //@siclog "Steps 10Ea1" siclog@
        // @sic R5-223436 sic@
        //Check: Does the UE transmit an RRCSetupREQ message with establishmentCause of mo_Data?
        [] SRB.receive(car_NR_SRB0_RrcPdu_IND(nr_Cell12, cr_38508_RRCSetupRequest(mo_Data, ?))) //@sic R5s221172 sic@
        {
           //@siclog "Steps 10Ea2" siclog@ // @sic R5-223436 sic@
           t_WaitTime.stop;
           f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 10Ea1");
        }
        //@siclog "Steps 10Eb1" siclog@ // @sic R5-223436 sic@
        [] t_WaitTime.timeout {
            //@siclog "Steps 10Eb2" siclog@ // @sic R5-223436 sic@
            f_UT_RequestPDUSession(UT, f_NR5GC_MobileInfo_GetNoOfSessions()+1, tsc_DNNValue);
            t_Wait.start(10.0);
            alt {
            //@siclog "Steps 10Eb3" siclog@ // @sic R5-223436 sic@
                [] SRB.receive(car_NR_SRB0_RrcPdu_IND(nr_Cell12, cr_38508_RRCSetupRequest(mo_Data, ?)))
                {
                    f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 10Eb3");
                }
                []t_Wait.timeout {
                    f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 10Eb3");
                }
            }
        }
    }
    //@siclog "Steps 10Fa1-10Fa8" siclog@
    // @sic R5-223436 sic@
    f_NR_ULGrantConfiguration_Start(nr_Cell12, -, -, -, cs_NR_UplinkTimeAlignment_Stop, -, -, tsc_NoCnfReq ); // @sic R5s221172 sic@
    f_NR_RRCSetup_Def (nr_Cell12);
    v_ReceivedMsg := f_NR_RRCSetupComplete_Def(nr_Cell12,
                                          cr_38508_RRCSetupComplete(tsc_NR_RRC_TI_Def, ?,- , *, *),
                                          tsc_SHT_IntegrityProtected
                                          );
    if (not f_Check_NG_ServiceReqMsg (v_ReceivedMsg.Pdu, cs_NAS_KeySetIdentifier_lv(f_NR_SecurityKSIamf_Get (), tsc_NasKsi_NativeSecurityContext), v_S_TMSI, '0000'B)) { //R5s230103
      f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Service Request Message Failed");
    }
    f_NR5GC_RRC_ConnectedState3N_Steps5_8(nr_Cell12, v_ReceivedMsg); // @sic R5s221172 sic@
    f_NR5GC_PDUSessionEstablishment (nr_Cell12,-,tsc_NR_RbId_SRB1);

    //@siclog "Steps 10H" siclog@ // @sic R5-223436 sic@
    //The UE is switched off by executing generic procedure in Table 4.9.6.3-1 in TS 38.508-1 [4]
    f_NR_UE_DeRegisterOnSwitchOff(nr_Cell12, STATE_CONNECTED_3A);
    
    f_NR5GC_MobileInfo_Init();
    //@siclog "Steps 11A" siclog@
    // @sic R5-223436 sic@
    f_NR_SetCellPowerList(v_CellPowerList_T2);
    //@siclog "Step 11B" siclog@ // @sic R5-223436 sic@
    //The UE is Switched ON. The generic test procedure in TS 38.508-1 [4] Table 4.5.2.2-2 indicate that the UE performs registration on NR Cell 1.
    f_NR5GC_SwitchOnAndResetIMS();
    f_NR5GC_RRC_IdleState1N_Def(nr_Cell1);
    //@siclog "Step 11C" siclog@ // @sic R5-223436 sic@
    f_NR_SetCellPowerList(v_CellPowerList_T1);
    //@siclog "Step 11D- 11F" siclog@ // @sic R5-223436 sic@
    //Steps 2-4 of Table 4.5.2.2-2 of the generic procedure in TS 38.508-1 [4] are performed on NR Cell 12. (Note 2)
    v_ReceivedMsg := f_NR_RRC_ConnEst_DefWithNas (nr_Cell12, tsc_NR_RRC_TI_Def, ?, (tsc_SHT_NoSecurityProtection, tsc_SHT_IntegrityProtected));
    
    if (f_Check_NG_RegistrationReqMsg (v_GMM_MobilityInfo, v_ReceivedMsg.Pdu, Initial_Secure)) {
    //@siclog "Steps 11Fa1-11Fa16a1" siclog@
    //IF 5GS registration type is set as Initial Registration in step 11F, THEN Steps 5 to 20a1of the generic test procedure in TS 38.508-1 Table 4.5.2.2-2 are performed on NR Cell 12.
    //f_NR5GC_RRC_Idle_Steps5_20 (v_GMM_MobilityInfo, nr_Cell12, v_ReceivedMsg, TESTMode_OFF, -, Initial_Secure);
    f_NR5GC_RRC_Idle_Steps5_13 (v_GMM_MobilityInfo, nr_Cell12, v_ReceivedMsg, v_PLMN, TESTMode_OFF, Initial_Secure, NOT_SUPPORTED);
    v_MsgToSend := f_Get_NG_RegistrationAcceptMsg (Initial_Secure,
                                                   v_GMM_MobilityInfo,
                                                   v_GUTIToSend,
                                                   v_TAIList,
                                                   omit,
                                                   -, -, -,
                                                   cds_NG_NetworkFeatureSupport_mpsi_TC1135);
    SRB.send(cas_NR_SRB_NasPdu_REQ(nr_Cell12, tsc_NR_RbId_SRB1, -, cs_NG_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered, v_MsgToSend)));
    // Step 15
    f_NR5GC_Registration_Complete(nr_Cell12);
    } else{
        if (f_Check_NG_RegistrationReqMsg (v_GMM_MobilityInfo, v_ReceivedMsg.Pdu, Mobility)) {
        //@siclog "Steps 11Fb1-11Fb3a1" siclog@
        //IF 5GS registration type is set as Mobility Registration in step 211F, THEN Steps 4 to 6a1 of the generic test procedure in TS 38.508-1 Table 4.9.5.2.2-1 are performed on NR Cell 12.
        // @sic R5-223436 sic@
          //f_NR5GC_MobilityRegistration_Steps4_6(nr_Cell12, v_GMM_MobilityInfo, -, -, -);
          v_MsgToSend := f_Get_NG_RegistrationAcceptMsg (Mobility,
                                                   v_GMM_MobilityInfo,
                                                   v_GUTIToSend,
                                                   v_TAIList,
                                                   omit,
                                                   -, -, -,
                                                   cds_NG_NetworkFeatureSupport_mpsi_TC1135);
         SRB.send(cas_NR_SRB_NasPdu_REQ(nr_Cell12, tsc_NR_RbId_SRB1, -, cs_NG_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered, v_MsgToSend)));
         f_NR5GC_Registration_Complete(nr_Cell12); // No SRB2 so must be on SRB1
        }
    }
    
    
    f_NR_RRCRelease(nr_Cell12);
    //@siclog "Step 12 siclog@
    //The SS transmits a Paging message including a matched identity (correct fullI-RNTI).
    f_NR_Paging_Def(nr_Cell12, RRC_IDLE);

    //@siclog "Step 14 siclog@
    t_WaitTime.start(v_TimerValue1);
    alt {
        //Check: Does the UE transmit an RRCSetupREQ message with establishmentCause of mps-PriorityAccess?
        [] SRB.receive(car_NR_SRB0_RrcPdu_IND(nr_Cell12, cr_38508_RRCSetupRequest(mps_PriorityAccess, ?)))
        {
           t_WaitTime.stop;
           f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 14");
        }
        [] t_WaitTime.timeout {
           f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 14");
        }
    }
…
…
…
}




After Change
	  function fl_TC_11_3_5_TestBody() runs on NR5GC_PTC
  {
…
…
…
    //@siclog "Steps 2Ca1-2Ca15" siclog@
    //IF 5GS registration type is set as Initial Registration in step 2B, THEN Steps 5 to 19 of the generic test procedure in TS 38.508-1 Table 4.5.2.2-2 are performed without RRC release.on NR Cell 12
    // @sic R5-223436 sic@
    
    select (v_ReceivedMsg.Pdu.Msg.registration_Request.registrationType.registrationType) {
        case (tsc_NG_RegistrationInitial) {
              //@siclog "Steps 2Ca1-2Ca15" siclog@
              //Perform steps 5 to 19 of the generic test procedure in TS 38.508-1 Table 4.5.2.2-2 
              if (not f_Check_NG_RegistrationReqMsg (v_GMM_MobilityInfo, v_ReceivedMsg.Pdu, Initial_Secure))
              {
                f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Registration Request Message Failed");
              }
              f_NR5GC_RRC_Idle_Steps5_20 (v_GMM_MobilityInfo, nr_Cell12, v_ReceivedMsg, TESTMode_OFF, noRrcConnectionRelease, Initial_Secure);
        }
        case (tsc_NG_RegistrationMobility) {
              //@siclog "2Cb1-2Cb2" siclog@
              //Perform steps 4 to 5 of the generic test procedure in TS 38.508-1 Table 4.9.5.2.2-1 
             if (not f_Check_NG_RegistrationReqMsg (v_GMM_MobilityInfo, v_ReceivedMsg.Pdu, Mobility))
              {
                f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Registration Request Message Failed");
              }
             f_NR5GC_MobilityRegistration_Steps4_6(nr_Cell12, v_GMM_MobilityInfo, -, -, noRrcConnectionRelease);
        }
        case else {
          f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Registration type incorrect");
        }
    } 

    //@siclog "Steps 3 - 10" siclog@
    //Void @sic R5-223436 sic@
    //@siclog "Steps 10A" siclog@
    // @sic R5-223436 sic@
    f_NR_RRCRelease(nr_Cell12);
    //@siclog "Steps 10B" siclog@
    // @sic R5-223436 sic@
    f_UT_RequestPDUSession(UT, f_NR5GC_MobileInfo_GetNoOfSessions()+1, tsc_DNNValue);
    //@siclog "Steps 10C" siclog@
    // @sic R5-223436 sic@
    t_WaitTime.start(21.0);
    alt {
        //Check: Does the UE transmit an RRCSetupRequest message within 21s on NR Cell 12?
        [] SRB.receive(car_NR_SRB0_RrcPdu_IND(nr_Cell12, cr_38508_RRCSetupRequest(?, ?)))
        {
           t_WaitTime.stop;
           f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 10C");
        }
        [] t_WaitTime.timeout {
           f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 10C");
        }
    }
    //@siclog "Steps 10D" siclog@
    // @sic R5-223436 sic@
    //SS changes SIB1 of NR Cell 12 according to Table 11.3.5.3.3-2 and sends Short Message on PDCCH using P-RNTI.
    f_NR_CellInfo_SetSIB1_UacBarringInfo(nr_Cell12, omit);
    f_NR_ModifySysinfo(nr_Cell12);
    //@siclog "Steps 10E" siclog@
    // @sic R5-223436 sic@
    t_WaitTime.start(28.0);
    alt {
        //@siclog "Steps 10Ea1" siclog@
        // @sic R5-223436 sic@
        //Check: Does the UE transmit an RRCSetupREQ message with establishmentCause of mo_Data?
        [] SRB.receive(car_NR_SRB0_RrcPdu_IND(nr_Cell12, cr_38508_RRCSetupRequest(mo_Data, ?))) //@sic R5s221172 sic@
        {
           //@siclog "Steps 10Ea2" siclog@ // @sic R5-223436 sic@
           t_WaitTime.stop;
           f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 10Ea1");
        }
        //@siclog "Steps 10Eb1" siclog@ // @sic R5-223436 sic@
        [] t_WaitTime.timeout {
            //@siclog "Steps 10Eb2" siclog@ // @sic R5-223436 sic@
            f_UT_RequestPDUSession(UT, f_NR5GC_MobileInfo_GetNoOfSessions()+1, tsc_DNNValue);
            t_Wait.start(10.0);
            alt {
            //@siclog "Steps 10Eb3" siclog@ // @sic R5-223436 sic@
                [] SRB.receive(car_NR_SRB0_RrcPdu_IND(nr_Cell12, cr_38508_RRCSetupRequest(mo_Data, ?)))
                {
                    t_Wait.stop;
                    f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 10Eb3");
                }
                []t_Wait.timeout {
                    f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 10Eb3");
                }
            }
        }
    }
    //@siclog "Steps 10Fa1-10Fa8" siclog@
    // @sic R5-223436 sic@
    f_NR_ULGrantConfiguration_Start(nr_Cell12, -, -, -, cs_NR_UplinkTimeAlignment_Stop, -, -, tsc_NoCnfReq ); // @sic R5s221172 sic@
    f_NR_RRCSetup_Def (nr_Cell12);
    v_ReceivedMsg := f_NR_RRCSetupComplete_Def(nr_Cell12,
                                          cr_38508_RRCSetupComplete(tsc_NR_RRC_TI_Def, ?,- , *, *),
                                          tsc_SHT_IntegrityProtected
                                          );
    if (not f_Check_NG_ServiceReqMsg (v_ReceivedMsg.Pdu, cs_NAS_KeySetIdentifier_lv(f_NR_SecurityKSIamf_Get (), tsc_NasKsi_NativeSecurityContext), v_S_TMSI, '0000'B)) { //R5s230103
      f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Service Request Message Failed");
    }
    f_NR5GC_RRC_ConnectedState3N_Steps5_8(nr_Cell12, v_ReceivedMsg); // @sic R5s221172 sic@
    f_NR5GC_PDUSessionEstablishment (nr_Cell12,-,tsc_NR_RbId_SRB2);

    //@siclog "Steps 10H" siclog@ // @sic R5-223436 sic@
    //The UE is switched off by executing generic procedure in Table 4.9.6.3-1 in TS 38.508-1 [4]
    f_NR_UE_DeRegisterOnSwitchOff(nr_Cell12, STATE_CONNECTED_3A);
    // Tell IMS PTC to stop waiting for De-register
    f_IMS_IPCAN_SendCoOrdMsg(IMS[tsc_Index_IMS1]);
    
    f_NR5GC_MobileInfo_Init();
    //@siclog "Steps 11A" siclog@
    // @sic R5-223436 sic@
    f_NR_SetCellPowerList(v_CellPowerList_T2);
    //@siclog "Step 11B" siclog@ // @sic R5-223436 sic@
    //The UE is Switched ON. The generic test procedure in TS 38.508-1 [4] Table 4.5.2.2-2 indicate that the UE performs registration on NR Cell 1.
    f_NR5GC_SwitchOnAndResetIMS();
    f_NR5GC_RRC_IdleState1N_Def(nr_Cell1);
    //@siclog "Step 11C" siclog@ // @sic R5-223436 sic@
    f_NR_SetCellPowerList(v_CellPowerList_T1);
    //@siclog "Step 11D- 11F" siclog@ // @sic R5-223436 sic@
    //Steps 2-4 of Table 4.5.2.2-2 of the generic procedure in TS 38.508-1 [4] are performed on NR Cell 12. (Note 2)
    v_ReceivedMsg := f_NR_RRC_ConnEst_DefWithNas (nr_Cell12, tsc_NR_RRC_TI_Def, ?, (tsc_SHT_NoSecurityProtection, tsc_SHT_IntegrityProtected));
    
    
    select (v_ReceivedMsg.Pdu.Msg.registration_Request.registrationType.registrationType) {
        case (tsc_NG_RegistrationInitial) {
              //Perform steps 5 to 19 of the generic test procedure in TS 38.508-1 Table 4.5.2.2-2 
              if (not f_Check_NG_RegistrationReqMsg (v_GMM_MobilityInfo, v_ReceivedMsg.Pdu, Initial_Secure))
              {
                f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Registration Request Message Failed");
              }
					    //@siclog "Steps 11Fa1-11Fa16a1" siclog@
					    //IF 5GS registration type is set as Initial Registration in step 11F, THEN Steps 5 to 20a1of the generic test procedure in TS 38.508-1 Table 4.5.2.2-2 are performed on NR Cell 12.
					    //f_NR5GC_RRC_Idle_Steps5_20 (v_GMM_MobilityInfo, nr_Cell12, v_ReceivedMsg, TESTMode_OFF, -, Initial_Secure);
					    f_NR5GC_RRC_Idle_Steps5_13 (v_GMM_MobilityInfo, nr_Cell12, v_ReceivedMsg, v_PLMN, TESTMode_OFF, Initial_Secure, NOT_SUPPORTED);
					    v_MsgToSend := f_Get_NG_RegistrationAcceptMsg (Initial_Secure,
					                                                   v_GMM_MobilityInfo,
					                                                   v_GUTIToSend,
					                                                   v_TAIList,
					                                                   omit,
					                                                   -, -, -,
					                                                   cds_NG_NetworkFeatureSupport_mpsi_TC1135);
					    SRB.send(cas_NR_SRB_NasPdu_REQ(nr_Cell12, tsc_NR_RbId_SRB1, -, cs_NG_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered, v_MsgToSend)));
					    // Step 15
					    f_NR5GC_Registration_Complete(nr_Cell12);
        }
        case (tsc_NG_RegistrationMobility) {
              //Perform steps 4 to 5 of the generic test procedure in TS 38.508-1 Table 4.9.5.2.2-1 
             if (not f_Check_NG_RegistrationReqMsg (v_GMM_MobilityInfo, v_ReceivedMsg.Pdu, Mobility))
              {
                f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Registration Request Message Failed");
              }
			        //@siclog "Steps 11Fb1-11Fb3a1" siclog@
			        //IF 5GS registration type is set as Mobility Registration in step 211F, THEN Steps 4 to 6a1 of the generic test procedure in TS 38.508-1 Table 4.9.5.2.2-1 are performed on NR Cell 12.
			        // @sic R5-223436 sic@
			          //f_NR5GC_MobilityRegistration_Steps4_6(nr_Cell12, v_GMM_MobilityInfo, -, -, -);
			          v_MsgToSend := f_Get_NG_RegistrationAcceptMsg (Mobility,
			                                                   v_GMM_MobilityInfo,
			                                                   v_GUTIToSend,
			                                                   v_TAIList,
			                                                   omit,
			                                                   -, -, -,
			                                                   cds_NG_NetworkFeatureSupport_mpsi_TC1135);
			         SRB.send(cas_NR_SRB_NasPdu_REQ(nr_Cell12, tsc_NR_RbId_SRB1, -, cs_NG_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered, v_MsgToSend)));
			         f_NR5GC_Registration_Complete(nr_Cell12); // No SRB2 so must be on SRB1
        }
        case else {
          f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Registration type incorrect");
        }
    }//anupams@11.3.5     
    
    
    f_NR_RRCRelease(nr_Cell12);
    //@siclog "Step 12 siclog@
    //The SS transmits a Paging message including a matched identity (correct fullI-RNTI).
    f_NR_Paging_Def(nr_Cell12, RRC_IDLE);

    //@siclog "Step 14 siclog@
    t_WaitTime.start(v_TimerValue1);
    alt {
        //Check: Does the UE transmit an RRCSetupREQ message with establishmentCause of mps-PriorityAccess?
        [] SRB.receive(car_NR_SRB0_RrcPdu_IND(nr_Cell12, cr_38508_RRCSetupRequest(mps_PriorityAccess, ?)))
        {
           t_WaitTime.stop;
           f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 14");
        }
        [] t_WaitTime.timeout {
           f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 14");
        }
    }
…
…
…
  }




2. [bookmark: _Toc128145558]Change 3
	Function name
	f_TC_11_3_5_NR5GC_IMS1

	Reason for change
	The coordination trigger on the NR PTC expects the trigger message to be “IMS Call Release Msgs Sent”, while the current implementation sends a blank string which causes a failure.

	Summary of change
	Modified the coordination trigger message to “IMS Call Release Msgs Sent”

	TTCN module
	UAC_NR5GC_IMS.ttcn

	MCC160 Comment
	



Before Change
	  function f_TC_11_3_5_NR5GC_IMS1()runs on IMS_PTC
  {
    f_IMS_PTC_Init(PDN_1);

    f_IMS_IPCAN_SendCoOrdMsg(IPCAN);
    f_IMS_Registration();
    f_IMS_IPCAN_SendCoOrdMsg(IPCAN);
    
    f_IMS_IpcanReleaseWithOptionalDeregistration();
    f_IMS_GlobalRelease();
    f_IMS_IPCAN_SendCoOrdMsg(IPCAN);
    f_IMS_Registration();
    f_IMS_IPCAN_SendCoOrdMsg(IPCAN);
    
    // MO Video Call
    f_IMS_NR5GC_MOVideoCall();
    
    // Call Release MT
    f_IMS_CallReleaseMT(f_SIP_BuildSipUri_TX(px_IMS_CalleeContactUri));
    //Tell NR that call is released
    f_IMS_IPCAN_SendCoOrdMsg(IPCAN);
    f_IMS_IpcanReleaseWithOptionalDeregistration();
    
  }    return "";
  }



After Change
	   function f_TC_11_3_5_NR5GC_IMS1()runs on IMS_PTC
  {
    f_IMS_PTC_Init(PDN_1);

    f_IMS_IPCAN_SendCoOrdMsg(IPCAN);
    f_IMS_Registration();
    f_IMS_IPCAN_SendCoOrdMsg(IPCAN);
    
    f_IMS_IpcanReleaseWithOptionalDeregistration();
    f_IMS_GlobalRelease();
    f_IMS_IPCAN_SendCoOrdMsg(IPCAN);
    f_IMS_Registration();
    f_IMS_IPCAN_SendCoOrdMsg(IPCAN);
    
    // MO Video Call
    f_IMS_NR5GC_MOVideoCall();
    
    // Call Release MT
    f_IMS_CallReleaseMT(f_SIP_BuildSipUri_TX(px_IMS_CalleeContactUri));
    //Tell NR that call is released
    //f_IMS_IPCAN_SendCoOrdMsg(IPCAN);
    f_IMS_IPCAN_SendCoOrdMsg(IPCAN, cms_IMS_IPCAN_Trigger("IMS Call Release Msgs Sent"));
    f_IMS_IpcanReleaseWithOptionalDeregistration();
    
  } 




2. [bookmark: _Toc128145559]Change 4
	Function name
	f_IMS_NR5GC_MOVideoCall_Step2ToEnd

	Reason for change
	1. The PRACK message contents from TS 34.229-1 section A.2.4 does not mandate the UE to include sdp body.
2. The SIP UPDATE procedure can happen in parallel to receiving the PDU MODIFICATION COMPLETE message from the UE. This is inline with the MO Video call steps defined in TS 34.229-5.

	Summary of change
	1. Removed the argument from function f_IMS_MOCallSetup_ReceivePRACK_Send200OK() to expect the SDP body (and streamline it with the TS 34.229-5 expectation as well).
2. Moved the coordination message wait trigger after the SIP UPDATE procedure

	TTCN module
	IMS_NR5GC_Functions.ttcn

	MCC160 Comment
	



Before Change
	  function f_IMS_NR5GC_MOVideoCall_Step2ToEnd(IMS_InviteRequestWithSdp_Type p_InviteRequestWithSdp) runs on IMS_PTC
  {
    var IMS_InviteRequestWithSdp_Type v_InviteRequestWithSdp := p_InviteRequestWithSdp;
    var IMS_ResponseWithSdp_Type v_ResponseWithSdp;
    var template (present) SDP_Message v_UpdatedSdpOffer;
    
    // preconditions handled within function
    v_ResponseWithSdp := f_IMS_MOCallSetup_Video_Step3_4(SetupVideoCall, v_InviteRequestWithSdp, EVS_16000, H265_90000);
    v_UpdatedSdpOffer := f_IMS_BuildSDP_UpdatedOfferAudioVideo_RX(SetupVideoCall, v_InviteRequestWithSdp.SdpOffer, EVS_16000, H265_90000);
    f_IMS_MOCallSetup_ReceivePRACK_Send200OK(v_InviteRequestWithSdp, v_ResponseWithSdp.Response.msgHeader, -, v_UpdatedSdpOffer);     // A.15.1 Step 4 - 5    PRACK, 200 OK

    f_IMS_IPCAN_SendCoOrdMsg(IPCAN, cms_IMS_IPCAN_Trigger("Add Video Bearer")); // @sic R5-225362 sic@
    f_IMS_IPCAN_WaitForTrigger(IPCAN, "Video Bearer Added");  // @sic R5-225362 sic@
    if (pc_Preconditions) {
      f_IMS_MOCallSetup_NR5GC_Video_Step6_7(SetupVideoCall, v_InviteRequestWithSdp);            // A.15.1 Step 6 - 7    UPDATE, 200 OK
      f_IMS_MOCallSetup_Send180_Reliable(v_InviteRequestWithSdp);                               // A.15.1 Step 8 - 10   180 RInging, PRACK, 200 OK
    } else {
      f_IMS_MOCallSetup_NR5GC_A_15_2_Step6_8(v_InviteRequestWithSdp);
    }
    f_IMS_MOCallSetup_Send200OK_ReceiveACK(v_InviteRequestWithSdp);                             // A.15.1 Step 11 - 12  200 OK, ACK
    // Call established
    f_IMS_IPCAN_SendCoOrdMsg(IPCAN, cms_IMS_IPCAN_Trigger("Video Call Established"));           // @sic R5-225362 sic@
  }  



After Change
	  function f_IMS_NR5GC_MOVideoCall_Step2ToEnd(IMS_InviteRequestWithSdp_Type p_InviteRequestWithSdp) runs on IMS_PTC
  {
    var IMS_InviteRequestWithSdp_Type v_InviteRequestWithSdp := p_InviteRequestWithSdp;
    var IMS_ResponseWithSdp_Type v_ResponseWithSdp;
    var template (present) SDP_Message v_UpdatedSdpOffer;
    
    // preconditions handled within function
    v_ResponseWithSdp := f_IMS_MOCallSetup_Video_Step3_4(SetupVideoCall, v_InviteRequestWithSdp, EVS_16000, H265_90000);
    v_UpdatedSdpOffer := f_IMS_BuildSDP_UpdatedOfferAudioVideo_RX(SetupVideoCall, v_InviteRequestWithSdp.SdpOffer, EVS_16000, H265_90000);
    f_IMS_MOCallSetup_ReceivePRACK_Send200OK(v_InviteRequestWithSdp, v_ResponseWithSdp.Response.msgHeader);     // A.15.1 Step 4 - 5    PRACK, 200 OK //REMOVED v_UpdatedSdpOffe

    f_IMS_IPCAN_SendCoOrdMsg(IPCAN, cms_IMS_IPCAN_Trigger("Add Video Bearer")); // @sic R5-225362 sic@
    if (pc_Preconditions) {
      f_IMS_MOCallSetup_NR5GC_Video_Step6_7(SetupVideoCall, v_InviteRequestWithSdp);            // A.15.1 Step 6 - 7    UPDATE, 200 OK
      f_IMS_MOCallSetup_Send180_Reliable(v_InviteRequestWithSdp);                               // A.15.1 Step 8 - 10   180 RInging, PRACK, 200 OK
    } else {
      f_IMS_MOCallSetup_NR5GC_A_15_2_Step6_8(v_InviteRequestWithSdp);
    }
    f_IMS_MOCallSetup_Send200OK_ReceiveACK(v_InviteRequestWithSdp);                             // A.15.1 Step 11 - 12  200 OK, ACK
    f_IMS_IPCAN_WaitForTrigger(IPCAN, "Video Bearer Added");  // @sic R5-225362 sic@ //MOVED HERE
    // Call established
    f_IMS_IPCAN_SendCoOrdMsg(IPCAN, cms_IMS_IPCAN_Trigger("Video Call Established"));           // @sic R5-225362 sic@
  }






